Maple-assisted proof of empirical formula for A263330

Let the "state" of a 3-column array of 0..2 be the values mod 11 of the columns considered as base-3 numbers. Thus there are $11^3 = 1331$ possible states to consider. The possible rows are the base-3 numbers of up to 3 digits that are divisible by 11, namely $000_3 = 0$, $102_3 = 11$ and $211_3 = 22$. Let T be the 1331×1331 matrix such that T_{ij} is the number of possible rows r such that an array in state number i atop row r will produce an array in state number j. Thus if the first array's state is (x, y, z) and the row is (t, u, v), the new array has state $(3x + t, 3y + u, 3z + v)$ (with operations done mod 11). Each row of T has three nonzero entries. The following code produces the matrix T.

```
> States:= [seq(seq(seq([x,y,z],z=0..10),y=0..10),x=0..10)]:
  T:= Matrix(1331,1331,storage=sparse):
  for i from 1 to 1331 do
     for y in [[0,0,0],[1,0,2],[2,1,1]] do
       z:= 3*States[i]+y mod 11;
       j:= 121*z[1]+11*z[2]+z[3] + 1;
       T[i,j]:= 1;
  od od:
```

Considering an array of 0 rows as being in state $(0, 0, 0)$, which is the first in our enumeration, $a(n)$ should be $e_1^T T^{n+2} e_1$ where e_1 is the unit vector with 1 in the first position. To confirm, we compute the first few entries. For future use, $U_j = T^j e_1$.

```
> U[0]:= Vector([1,0$1330]):
  for j from 1 to 30 do U[j]:= T . U[j-1] od:
  seq(U[n+2][1],n=1..27);
```

1, 1, 3, 9, 23, 51, 133, 399, 1321, 4129, 12457, 38059, 114177, 349017, 1056893, 3188001, 9592717, 28778151, 86390521, 259279771, 778196971, 2334819901, 7004459703, 21010336329, 63026931713, 189077563341, 567218511883 **(1)**

Now the empirical formula is

```
> Emp:= a(n) = 3*a(n-1) +16*a(n-5) -48*a(n-6) -505*a(n-10) +1515*a
  (n-11) -5105*a(n-15) +15315*a(n-16) -10114*a(n-20) +30342*a(n-21)
  +15709*a(n-25) -47127*a(n-26);
```

$$Emp := a(n) = 3\,a(n-1) + 16\,a(n-5) - 48\,a(n-6) - 505\,a(n-10) + 1515\,a(n-11) - 5105\,a(n-15) + 15315\,a(n-16) - 10114\,a(n-20) + 30342\,a(n-21) + 15709\,a(n-25) - 47127\,a(n-26) \quad \textbf{(2)}$$

This corresponds to saying $e_1^T P(T)\, T^{n+2} e_1 = 0$ where P is the following polynomial.

```
> P:= t^26 - add(coeff(rhs(Emp),a(n-j))*t^(26-j),j=0..26);
```

$$P := t^{26} - 3\,t^{25} - 16\,t^{21} + 48\,t^{20} + 505\,t^{16} - 1515\,t^{15} + 5105\,t^{11} - 15315\,t^{10} + 10114\,t^{6} - 30342\,t^{5} - 15709\,t + 47127 \quad \textbf{(3)}$$

We compute $P(T)\, e_1$ using the previously computed values U_j, and verify that it is 0 :

```
> Q:= add(coeff(P,t,j)*U[j],j=0..26):
> LinearAlgebra:-Equal(Q,LinearAlgebra:-ZeroVector(1331));
```

true **(4)**

Thus we have $e_1^T P(T)\, T^{n+2} e_1 = e_1^T\, T^{n+2} P(T)\, e_1 = 0$. This completes the proof.